

EDITORALUTÉCIA

# PROGRAMAS DE JOGOS DE HORROR



Para o Spectrum, TRS-80, TRS-80 Color, Apple, MSX, PC, Commodore 64,

VIC 20 e todos os micros nacionais compatíveis.



Titulo original inglês  
**CREEPY COMPUTER GAMES**

Copyright © 1983 by Usborne Publishing Ltd.

Direitos de publicação exclusiva em língua portuguesa em todo o mundo  
adquiridos pela

**EDITORA LUTÉCIA LTDA.**

Rua Argentina 171 — 20921 Rio de Janeiro, RJ — Tel.: 580-3668  
que se reserva a propriedade literária desta tradução

---

Impresso no Brasil


**PEDIDOS PELO REEMBOLSO POSTAL**  
Caixa Postal 23.052 — Rio de Janeiro, RJ — 20922

Impresso por  
Gráfica Portinho Cavalcanti Ltda.  
Rua Santana, 136/138 (edifício próprio)  
Tel.: 224-7732 (PABX)  
Rio de Janeiro — RJ



# PROGRAMAS DE JOGOS DE HORROR

Tradução e Adaptação de Ronaldo Sergio de Biasi



Sumário
3 Pesadelo
4 O Mago dos Números
6 Aspirador de Fantasma
8 Mulher-Aranha
10 Cemitério
12 Casa Maluca
14 Labirinto Assombrado
16 Sessão Espírita
17 Respostas dos Problemas

EDITORAL LUTÉCIA



# Introdução

Embora todos os micros nacionais utilizem a linguagem BASIC, existem vários dialetos desta linguagem usados em micros de diferentes marcas e modelos. Os micros nacionais podem ser divididos em seis "famílias": a família Spectrum (TK-85, TK-90X), a família TRS-80 (DGT-1000, CP-500 etc.), a família TRS-80 Color (CP-400), a família Apple (Micro Engenho, Ap-II etc.), a família MSX (Expert, Hot Bit) e a família PC (Ego, I-7000 etc.). Se você não souber a que família pertence o seu micro, poderá obter a informação em qualquer loja especializada.

Neste livro, a listagem principal de cada programa é para os micros da família TRS-80. As linhas que precisam ser mudadas para que o programa funcione em outros computadores estão indicadas por símbolos e as modificações aparecem no final de cada listagem. Os símbolos são os seguintes:

- família TRS-80 Color
- ▼ família Spectrum
- família Apple
- família MSX
- ◆ família PC
- ▲ Commodore 64 e VIC 20\*

Quando você vir o símbolo da família a que pertence o seu micro, procure no final da listagem uma linha com o mesmo símbolo e o mesmo número. Use-a no lugar da linha correspondente da listagem principal. Na página 17 há notas especiais para os usuários do TK-2000, do MSX e do PC.

\*Até o momento não têm equivalentes no Brasil.

O programa **Pesadelo** se baseia em uma idéia de Brendon Kavanagh. O **Mago dos Números**, **Aspirador de Fantasmas** e **Labirinto Assombrado** foram escritos por Colin Reynolds.

**Mulher-Aranha** foi escrito por Val Robinson, **Cemitério** por Alan Ramsey e **Casa Maluca** por Keith Campbell.

Ilustrações: Rob McCaig

Edição: Jenny Tyler

Edição dos programas: Chris Oxlade

## Como carregar e executar os programas

Escreva as linhas exatamente como na listagem, incluindo todos os espaços e sinais de pontuação. Depois de bater cada linha, verifique na tela se tudo está certo antes de apertar a tecla RETURN (ou a tecla equivalente para o seu micro). Para começar o jogo, entre com RUN. Em alguns jogos, as coisas acontecem muito depressa, de modo que é melhor ler atentamente as instruções antes de começar. Se o programa não funcionar corretamente, o melhor é listá-lo e comparar as instruções com as do livro, uma por uma, até descobrir onde está o erro.

## Experiências com os jogos

Este livro contém várias sugestões para modificar e ampliar os programas, mas não tenha medo de introduzir suas próprias alterações. Não há perigo de danificar o computador; se sua idéia não der certo, sempre poderá voltar ao programa original.

Você provavelmente terá vontade de mudar a velocidade de alguns jogos, especialmente depois que estiver familiarizado com eles. As instruções para fazer isso estão perto da listagem de cada programa.

Se o seu computador é capaz de gerar cores ou sons, consulte o manual para descobrir como funcionam as instruções correspondentes e experimente usá-las nos programas deste livro.



# Pesadelo

Você adormeceu quando estava programando o seu computador. De repente, ele começa a funcionar sozinho e passa a insultá-lo e a mostrar números na tela. Para vencer o computador, você precisa repetir os números que aparecem. A contagem inicial de 300 aumenta quando você acerta um número e diminui quando erra. Se chegar a 500, o computador será derrotado, mas se chegar a zero, você se tornará escravo do micro. É um pesadelo! Quem ganhará?

## Como funciona

10 LET FS=""	
▼ 20 DIM CS(5)	
30 LET S=300	Inicializa a contagem de pontos.
40 LET CS(1)="SEU BURRO!"	
50 LET CS(2)="NAO PODE COMIGO!"	
60 LET CS(3)="SOU MAIS ESPERTO!"	
70 LET CS(4)="ABAIXO OS HUMANOS!"	
80 LET CS(5)="ESTA ACERTANDO MUITO!"	Guarda os comentários na memória.
▲ 90 CLS	Apaga a tela.
◆▼▲● 100 LET N=INT(RND(0)*9)+1	Escolhe um número de 1 a 9 e o guarda em N.
110 PRINT TAB(5);N	
120 PRINT TAB(15);S	Imprime N e a contagem (S) em diferentes locais da tela.
◆▼▲● 130 IF RND(0)>0.5 THEN GOTO 160	
140 PRINT	Decide se vai imprimir mensagens.
150 PRINT CS(INT(S/100)+1)	
160 IF S<60 THEN PRINT "ESTA PERDIDO!"	
170 IF S>440 THEN PRINT "SOCORRO!"	Imprime mensagens que dependem da contagem de pontos.
▼ 180 FOR I=1 TO 100	
▲ 190 LET AS=INKEY\$	
200 IF AS<>" " THEN LET FS=AS	Verifica se você está apertando uma tecla; se está, guarda em FS.
210 NEXT I	
220 LET S=S-10	Diminui a contagem de 10.
230 IF VAL(FS)<>N THEN GOTO 250	Se você apertou a tecla errada, pula para 250 para não aumentar a contagem na linha 240.
240 LET S=S+10+N*2	
250 IF S<0 THEN GOTO 280	
260 IF S>500 THEN GOTO 300	Verifica se você ganhou ou perdeu o jogo.
270 GOTO 90	
280 PRINT "VOCE É MEU ESCRAVO!"	
290 STOP	
300 PRINT "VOCE GANHOU (DESTA VEZ)"	
310 STOP	

A listagem acima é para o TRS-80. Para os outros micros, faça as mudanças abaixo.

```

▼ 20 DIM CS(5,23)
● 90 HOME
▲ 90 PRINT CHR$(147)
▲● 100 LET N=INT(RND(1)*9)+1
◆▼ 100 LET N=INT(RND*9)+1
▲● 130 IF RND(1)>0.5 THEN GOTO 160
◆▼ 130 IF RND>0.5 THEN GOTO 160
▼ 180 FOR I=1 TO 50
● 190 AS="":IF PEEK(-16384)>127
    THEN GET AS
▲ 190 GET AS
    
```


A rapidez do jogo depende deste número. Você provavelmente terá que ajustá-lo de acordo com a velocidade do seu micro. Quanto maior o número, mais lento o jogo.

## Sugestão

A contagem de pontos é controlada pelas linhas 30, 220 e 240. Mude os números nessas linhas para ver o que acontece.

## Problema

Como você faria para o computador mostrar letras na tela em lugar de números?





# O Mago dos Números



O Mago dos Números submete todos os visitantes a este desafio. Ele faz os números de 1 a 9 aparecerem no ar e depois joga dois dados. Você tem que dar a ele dois dos números que apareceram no ar e cuja soma seja igual à soma dos números dos dois dados. Depois que você usa um número, ele desaparece e não pode ser usado de novo. Para ganhar, você precisa fazer todos os números desaparecerem antes de terminarem suas 8 tentativas (você ganha uma tentativa extra cada vez que os dados dão uma dobradinha). Você pode usar o zero como um dos números quantas vezes quiser. Se não puder obter a soma desejada, passe para a tentativa seguinte usando dois zeros. Veja quantas vezes consegue derrotar o Mago.



## Como o programa funciona

10 DIM A(10)	Cria uma matriz, A.
20 LET T=8	T é o número de tentativas.
30 FOR I=1 TO 10	Coloca 1 em todos os elementos de A.
40 LET A(I)=1	
50 NEXT I	
▲ 60 CLS	
70 PRINT "O MAGO DOS NUMEROS":PRINT	Coloca 1 em A(1), para que você possa usar 0 quantas vezes quiser.
80 LET A(1)=1	
90 LET V=0	V é o número de números que já foram usados.
100 FOR I=2 TO 10	Imprime os números. Se A(I) é zero, o número I-1 já foi usado.
110 IF A(I)<>0 THEN GOTO 150	
120 PRINT " "	
130 LET V=V+1	Imprime o número de tentativas que restam.
140 GOTO 160	
150 PRINT I-1;	
160 NEXT I	
170 PRINT	
180 IF T=1 THEN PRINT	
"ESTA E A ULTIMA TENTATIVA":GOTO 200	
190 PRINT "VOCE TEM ";T;" TENTATIVAS"	Escolhe os números dos dados e os imprime.
◆▼▲● 200 LET C=INT(RND(0)*6+1)	No caso de uma dobradinha, aumenta o número de tentativas que restam.
◆▼▲● 210 LET B=INT(RND(0)*6+1)	
220 PRINT "OS NUMEROS SAO ";C;" ";B	
230 PRINT "QUAIS SAO OS SEUS?"	Recebe os seus números e verifica se são válidos.
240 IF B=C THEN LET T=T+2	
250 INPUT N	
260 INPUT M	Se a soma está errada ou um dos números já foi usado, volta para outra tentativa.
270 IF M>9 OR N>9 THEN PRINT	
"GRANDE DEMAIS...TENDE DE NOVO"	
280 IF M>9 OR N>9 THEN GOTO 230	Se os números acabaram, você venceu.
290 IF M+N<>B+C OR A(N+1)=0 OR A(M+1)=0 THEN GOTO 340	
300 LET A(M+1)=0	Se os números estão certos, coloca 0 nos elementos correspondentes de A para mostrar que já foram usados.
310 LET A(N+1)=0	
320 IF V=8 AND (M<>0 OR N<>0) THEN GOTO 370	Se os números acabaram, você venceu.
330 IF V=7 AND M<>0 AND N<>0 THEN GOTO 370	
340 LET T=T-1:IF T>0 THEN GOTO 60	Diminui de 1 o número de tentativas e verifica se elas acabaram.
350 PRINT "O MAGO VENCEU"	
360 STOP	
370 PRINT "VOCE VENCEU"	
380 STOP	

A listagem acima é para o TRS-80. Para os outros micros, faça as modificações abaixo.

- 60 HOME
- ▲ 60 PRINT CHR\$(147)
- ▲● 200 LET C=INT(RND(1)\*6+1)
- ◆▼ 200 LET C=INT(RND\*6+1)
- ▲● 210 LET B=INT(RND(1)\*6+1)
- ◆▼ 210 LET B=INT(RND\*6+1)

Veja se consegue fazer o computador voltar automaticamente para um novo desafio e manter um registro do número de vezes que você e o Mago venceram.



# Aspirador de Fantasmas



Neste jogo, os fantasmas são números que passam na tela. Para capturá-los, você aciona o Aspirador de Fantasmas apertando a tecla X, mas para que ele funcione é preciso que esteja com o mesmo número que o fantasma. Para aumentar o número do aspirador, basta apertar a tecla M (quando o número chega a 9, volta de novo a 0). Quando você deixa de aspirar um fantasma, ele rouba uma das suas vidas (indicadas como barras no canto superior esquerdo da tela). Veja quantos fantasmas consegue aspirar.

## Como o programa funciona

10 LET S=0

20 LET Y=0

30 LET L=3

40 LET N=INT(RND(0)\*10)

50 LET I=1

60 CLS

70 PRINT "ASPIRADOR DE FANTASMAS":PRINT

80 FOR J=1 TO L

90 PRINT "/";

100 NEXT J

110 PRINT

120 PRINT TAB(1);N;TAB(18);":":Y

Inicializa a contagem e o número do aspirador.

Fixa o número de vidas.

Escolhe um número entre 0 e 9 para o fantasma.

Inicializa o contador da distância percorrida pelo fantasma.

Imprime na tela o número de vidas que restam.

Imprime o fantasma, a barreira (:) e o aspirador.

Este é um bom jogo para você acrescentar efeitos sonoros. Escolha um som para quando perder uma vida e outro para quando capturar um fantasma.



```

130 FOR K=1 TO 50
140 NEXT K
▲● 150 LET B$=INKEY$
160 IF B$="X" THEN GOTO 240
170 IF B$="M" THEN LET Y=Y+1
180 IF Y=10 THEN LET Y=0
190 LET I=I+1
200 IF I<18 THEN GOTO 60
210 LET L=L-1
220 IF L>0 THEN GOTO 40
230 GOTO 310
240 IF Y<>N THEN GOTO 190
250 PRINT "PEGOU UM"
260 PRINT "*****"
270 FOR K=1 TO 300
280 NEXT K
290 LET S=S+(18-I)
300 GOTO 40
▲● 310 CLS
320 PRINT "VOCE FEZ"
330 PRINT S; "PONTOS"
340 PRINT
350 PRINT "DE NOVO? (S/N)"
360 INPUT A$
370 IF A$="S" THEN RUN
380 IF A$="N" THEN STOP

```

Você pode mudar a velocidade deste jogo mudando o número da linha 130; quanto menor o número, mais rápido o jogo. Também pode tornar o jogo mais difícil diminuindo o número que aparece nas linhas 120 e 200.



Verifica se você está apertando uma tecla. Se é X, vai para 240; se é M, aumenta o seu número de 1.

Faz seu número voltar a zero quando ele passa de 9.

Incrementa o contador de distância. Se o fantasma ainda não chegou à barreira nem foi aspirado, volta para imprimir o fantasma na nova posição.

Se o fantasma chegou à barreira, você perde uma vida.

Se você ainda tem vidas, volta para sortear outro fantasma.

Verifica se o seu número é igual ao do fantasma; se não é, continua a incrementar o contador de distância.

A listagem acima é para o TRS-80. Para os outros micros, faça as mudanças abaixo.

```

▲■● 40 LET N=INT(RND(1)*10)
◆▼ 40 LET N=INT(RND*10)
● 60,310 HOME
▲ 60,310 PRINT CHR$(147)
● 150 LET B$="":IF
    PEEK(-16384))127
    THEN GET B$
▲150 GET B$

```



# Mulher-Aranha

Socorro! É a Mulher-Aranha!

Felizmente, hoje ela está de bom humor. Se você conseguir adivinhar a letra em que está pensando, ela o deixará ir. Se não, será transformado em mosca. Para descobrir a letra, dê uma palavra à Mulher-Aranha e ela lhe dirá se contém ou não a letra.





```

10 LET G=0
▲● 20 CLS
◆▼▲● 30 LET T=INT(RND(0)*26+65)
40 LET TS=CHR$(T)
50 PRINT "A MULHER-ARANHA"
60 PRINT "JA ESCOLHEU UMA LETRA"
70 PRINT
80 PRINT "DIGA UMA PALAVRA"
90 PRINT
100 PRINT
110 INPUT WS
120 LET G=G+1
130 L=LEN(WS)
140 IF L<4 OR L>8 THEN GOTO 70
150 LET F=0
160 FOR I=1 TO L
▼ 170 LET AS=MID$(WS,I,1)
180 IF AS=TS THEN F=1
190 NEXT I
200 IF F=1 THEN GOTO 280
210 PRINT
220 PRINT "NAO ESTA NESTA PALAVRA"
230 FOR A=1 TO 300
240 NEXT A
250 IF G>15 THEN GOTO 400
▲● 260 CLS
270 GOTO 70
280 PRINT "SIM - ESTA NESSA PALAVRA"
290 PRINT
300 PRINT "QUER DAR UM PALPITE(S/N)?"
310 INPUT RS
320 IF RS="N" THEN GOTO 250
330 PRINT
340 PRINT "QUAL O SEU PALPITE ?"
350 INPUT GS
360 IF GS<>TS THEN GOTO 410
370 PRINT "ESTA BEM - PODE IR"
380 PRINT "(DESTA VEZ)"
390 STOP
400 PRINT "TARDE DEMAIS"
410 PRINT "AGORA VOCE E UMA MOSCA"
420 STOP

```

A listagem acima é para o TRS-80. Para os outros micros, faça as mudanças abaixo.

```

● 20 HOME
▲ 20 PRINT CHR$(147)
▲■● 30 LET T=INT(RND(1)*26+65)
◆▼ 30 LET T=INT(RND*26+65)
▼ 170 LET AS=WS(I TO I)
● 260 HOME
▲ 260 PRINT CHR$(147)

```

## Como funciona

Escolhe um número, transforma-o em letra e guarda-o em TS.

Pede uma palavra.

Incrementa o número de tentativas.

Verifica se a palavra é grande ou pequena demais.

Verifica se a palavra contém a letra. Em caso afirmativo, vai para 280.

Imprime uma mensagem se a palavra não contém a letra.

"Loop" de espera para que você possa ver a mensagem.

Encerra o programa se suas tentativas acabaram.

Apaga a tela e volta para pedir outra palavra.

Imprime uma mensagem se a palavra contém a letra. Se você não quer dar um palpite, volta para pedir outra palavra.

Imprime mensagens diferentes se você acertou a letra ou não.

## Sugestões

Veja o que acontece se você mudar o 15 na linha 250 por um número menor. E se mudar o 4 e o 8 na linha 140?

## Problema

A Mulher-Aranha quer que você modifique o programa para fazer o seguinte: se a palavra contiver a letra escolhida, você terá direito a dois palpites, mas se o seu primeiro palpite estiver errado, perderá cinco tentativas.



# Cemitério

Você está perdido em um cemitério e tem até meia-noite para encontrar a saída. Esqueletos espreitam nas sombras e poderão matá-lo de susto se você se aproximar demais. Você pode cavar buracos (cinco por jogo) para mantê-los afastados, mas tome cuidado para não cair em um dos buracos que cavou.

Você não pode passar por cima dos túmulos (+) nem dos muros do cemitério (:). Os buracos são representados por (0), você é um (\*) e os esqueletos são (X). Veja se consegue escapar a tempo.

## Como o programa funciona

- ul style="list-style-type: none; padding-left: 0;">
- 50-90: Enche a matriz A de espaços vazios.
- 110: Fixa o número máximo de buracos.
- 120-170: Define os símbolos a serem usados.
- 180-260: Guarda os muros na matriz.
- 270-310: Escolhe a posição dos túmulos.
- 320-390: Fixa as posições iniciais do jogador e dos esqueletos.
- 400-420: Guarda os esqueletos na matriz.
- 440-620: Calcula a nova posição do jogador.
- 630-720: Verifica se há alguma coisa na nova posição. Se não há, move o jogador para lá.
- 750-830: Imprime vários finais para o jogo.
- 840-970: Imprime o cemitério.
- 1020-1070: Coloca um buraco na matriz se você assim o quer.
- 1080-1280: Move os esqueletos.

```

▲●10 CLS
20 PRINT "CEMITERIO"
30 DIM A(10,20)
40 DIM B(6)
50 FOR I=1 TO 10
60 FOR J=1 TO 20
▼70 LET A(I,J)=ASC(" ")
80 NEXT J
90 NEXT I
100 LET W=0
110 LET X=5
▼120 LET Y=ASC("*")
▼130 LET B=ASC("+")
▼140 LET C=ASC("0")
▼150 LET D=ASC(":")
▼160 LET E=ASC("X")
▼170 LET Z=ASC(" ")
180 FOR J=1 TO 10
190 LET A(J,1)=D
200 LET A(J,20)=D
210 NEXT J
220 FOR J=1 TO 20
230 LET A(1,J)=D
240 LET A(10,J)=D
250 NEXT J
260 LET A(9,20)=Z
270 FOR J=1 TO 20
♦▼▲●280 LET F=INT(RND(0)*7+2)
♦▼▲●290 LET G=INT(RND(0)*15+3)
300 LET A(F,G)=B
310 NEXT J
320 LET M=2
330 LET N=2
340 LET B(2)=19
350 LET B(4)=19
360 LET B(6)=19
370 LET B(1)=4
380 LET B(3)=3
390 LET B(5)=2
400 FOR J=1 TO 5 STEP 2
410 LET A(B(J),B(J+1))=E
420 NEXT J
430 GOSUB 840
440 PRINT "PARA ONDE VOCE
VAI (N,L,S OU 0)";
450 INPUT AS
460 IF AS="N" THEN GOTO 510
470 IF AS="L" THEN GOTO 540
480 IF AS="S" THEN GOTO 580
490 IF AS="0" THEN GOTO 610
500 GOTO 450

```



```

510 LET T=N-1
520 LET U=M
530 GOTO 630
540 LET T=N
550 LET U=M+1
560 IF A(T,U)=Z AND M=19 THEN
    GOTO 750
570 GOTO 630
580 LET T=N+1
590 LET U=M
600 GOTO 630
610 LET T=N
620 LET U=M-1
630 IF A(T,U)=Z THEN GOTO 690
640 IF A(T,U)=D OR A(T,U)=B
    THEN GOTO 670
650 IF A(T,U)=C THEN GOTO 780
660 IF A(T,U)=E THEN GOTO 810
670 PRINT "NAO PODE"
680 GOTO 440
690 LET A(N,M)=Z
700 LET N=T
710 LET M=U
720 LET A(N,M)=Y
730 GOSUB 1020
740 GOTO 400
750 PRINT "VOCE ESCAPOU"
760 PRINT "A NOTA PARA
    O SEU DESEMPENHO E ";
    INT((85-W)/60*(95+X));"%"
770 STOP
780 PRINT "VOCE CAIU EM UM"
790 PRINT "DE SEUS PROPRIOS
    BURACOS"
800 STOP
810 PRINT "ARGH! UM ESQUELETO"
820 PRINT "MATOU VOCE DE SUSTO"
830 STOP
840 LET A(N,M)=Y
▲● 850 CLS
860 LET R=N
870 LET S=M
880 LET W=W+1
890 IF W>60 THEN GOTO 980
900 PRINT "MOVIMENTO NUMERO ";W
910 FOR I=1 TO 10
920 FOR J=1 TO 20
930 PRINT CHR$(A(I,J));
940 NEXT J
950 PRINT
960 NEXT I

```


```

970 RETURN
980 PRINT "O RELOGIO BATEU"
990 PRINT "MEIA-NOITE"
1000 PRINT "AAAAAAHHHHHHH!!!!"
1010 STOP
1020 IF X=0 THEN GOTO 1080
1030 PRINT "QUER CAVAR UM
    BURACO(S OU N)";
1040 INPUT B$
1050 IF B$(">")="S" THEN GOTO 1080
1060 LET X=X-1
1070 LET A(R,S)=C
1080 FOR J=1 TO 5 STEP 2
1090 LET T=B(J)
1100 LET U=B(J+1)
1110 IF A(T,U+1)=Y THEN GOTO 810
1120 IF A(T,U-1)=Y THEN GOTO 810
1130 IF A(T-1,U)=Y THEN GOTO 810
1140 IF A(T+1,U)=Y THEN GOTO 810
1150 IF A$="O" THEN GOTO 1270
1160 IF A$="S" AND A(T+1,U)=Z
    THEN GOTO 1240
1170 IF A$="N" AND A(T-1,U)=Z
    THEN GOTO 1250
1180 IF A$="L" AND A(T,U+1)=Z
    AND M>U THEN GOTO 1210
1190 IF A$="L" AND A(T,U-1)=Z
    THEN GOTO 1220
1200 GOTO 1270
1210 LET B(J+1)=B(J+1)+2
1220 LET B(J+1)=B(J+1)-1
1230 GOTO 1260
1240 LET B(J)=B(J)+2
1250 LET B(J)=B(J)-1
1260 LET A(T,U)=Z
1270 NEXT J
1280 RETURN

```

A listagem acima é para o TRS-80. Para os outros micros, faça as mudanças abaixo.

- 10,850 HOME
- ▲ 10,850 PRINT CHR\$(147)
- ▼ 70,120-170 Substitua ASC por CODE
- ▲● 280,290 Substitua RND(0) por RND(1)
- ◆▼ 280,290 Substitua RND(0) por RND





# Casa Maluca

Você está preso em uma casa estranha onde tudo se move, até as paredes. Se pelo menos as portas ficassem alinhadas por um momento, você poderia correr para a liberdade! Você encontrou um painel que parece controlar alguns dos movimentos da casa. As teclas X e C fazem a porta mais próxima (na parte de cima da tela) mudar de direção. As teclas N e M têm o mesmo efeito sobre a porta mais distante. Parece não haver nenhum meio de controlar a porta do meio.

Enquanto você aperta nervosamente as teclas do painel, pode ouvir pegadas no corredor. O número no canto superior esquerdo do painel mostra que o fantasma está se aproximando. Se não conseguir fugir a tempo... aaaagghhhhh!

Se o seu computador tem som, veja se você consegue acrescentar efeitos sonoros aos passos e ao fim do jogo se você não conseguir escapar.

## Como funciona

Cria duas matrizes, uma para a posição (P) de cada porta e outra para a direção (S) em que está se movendo.

Cria uma fila de estrelas com a largura (W) da tela e a guarda em L\$.

Inicializa o contador dos passos do fantasma.

Fixa a direção do movimento das portas no início do jogo.

Escolhe as posições iniciais das três portas.

Imprime as paredes e as portas.

Imprime a contagem dos passos do fantasma.

Verifica se o fantasma alcançou você.

Verifica se você ganhou.

Recebe instruções do teclado e, se necessário, inverte o sentido do movimento das portas.

Movimenta a porta do centro a cada 25 passos.

Diminui a contagem de passos.

Calcula a nova posição das portas. Se uma porta chega à borda da tela, fica no mesmo lugar.

Volta para uma nova tentativa.

Imprime sua fuga e uma mensagem.

Imprime uma mensagem quando você perde.

```

10 DIM P(3)
20 DIM S(3)
30 CLEAR 130:CLS
40 PRINT "CASA MALUCA"
50 LET L$=""
60 LET W=63
70 FOR I=1 TO W
80 LET L$=L$+"*"
90 NEXT I
100 LET CT=240
110 LET S(1)=1
120 LET S(3)=-1
130 FOR I=1 TO 3
140 LET P(I)=RND(60)
150 NEXT I
160 CLS
170 FOR I=2 TO 12 STEP 5
180 PRINT$(I*64),L$
190 PRINT$(I*64+P((I+3)/5)), ">" "<"
200 NEXT I
210 PRINT$(0,CT)
220 IF CT<0 THEN GOTO 450
230 IF P(1)=P(2) AND P(2)=P(3) THEN GOTO 390
240 LET Z$=INKEY$
250 IF Z$="" THEN GOTO 300
260 IF Z$="X" THEN LET S(1)=-1
270 IF Z$="C" THEN LET S(1)=1
280 IF Z$="N" THEN LET S(3)=-1
290 IF Z$="M" THEN LET S(3)=1
300 IF CT/25=INT(CT/25) THEN LET P(2)=P(2)+RND(20)-10
310 LET CT=CT-1
320 LET P(1)=P(1)+S(1)
330 LET P(3)=P(3)+S(3)
340 FOR I=1 TO 3
350 IF P(I)<1 THEN LET P(I)=1
360 IF P(I)>(W-4) THEN LET P(I)=W-4
370 NEXT I
380 GOTO 170
390 LET L=P(1)+1
400 FOR I=1 TO 15
410 PRINT$(I*64+L),"M"
420 NEXT I
430 PRINT "VOCE ESCAPOU !!"
440 STOP
450 PRINT "TARDE DEMAIS !!"
460 PRINT "ARRRGGGHHHHH !!"
470 STOP
    
```



A listagem à esquerda é para o TRS-80. Para os outros micros, faça as mudanças indicadas nesta página.

● TRS-80 COLOR

```
30 CLS
60 LET W=31
140 LET P(1)=RND(28)
180 PRINT@ (I*32),L$
190 PRINT@ (I*32+P((I+3)/5)), " <"
410 PRINT@ (I*32+L),CHR$(34); 0 ;CHR$(34)
```

▼ SPECTRUM

```
30 CLS
60 LET W=31
140 LET P(1)=INT(RND*(W-4))+1
170 FOR I=5 TO 15 STEP 5
180 PRINT AT I,0;L$
190 PRINT AT I,P(1/5); " <"
210 PRINT AT 0,0;CT
300 IF CT/25=INT(CT/25) THEN LET
P(2)=P(2)+INT(RND*20)-10
410 PRINT AT I,L;"M"
```

● APPLE

```
30 HOME
60 LET W=39
140 LET P(1)=INT(RND(1)*(W-4))+1
160 HOME
170 FOR I=5 TO 15 STEP 5
180 VTAB(I)
185 PRINT L$
190 VTAB(I)
195 HTAB(P(1/5))
197 PRINT " <"
210 VTAB(1)=HTAB(1):PRINT CT
240 Z$=""
245 IF PEEK(-16384)>127 THEN GET Z$
300 IF CT/25=INT(CT/25) THEN LET
P(2)=P(2)+INT(RND(1)*20)-10
395 VTAB(1)
410 PRINT TAB(L)"M"
```

■ MSX

```
30 CLS
60 LET W=39
140 LET P(1)=INT(RND(1)*(W-4))+1
170 FOR I=5 TO 15 STEP 5
180 LOCATE 0,I:PRINT L$
190 LOCATE P(1/5),I:PRINT " <"
210 LOCATE 0,0:PRINT CT
300 IF CT/25=INT(CT/25) THEN LET
P(2)=P(2)+INT(RND(1)*20)-10
410 LOCATE L,I:PRINT "M"
```

◆ PC

```
30 CLS
60 LET W=79
140 LET P(1)=INT(RND*(W-4))+1
170 FOR I=5 TO 15 STEP 5
180 LOCATE 0,I:PRINT L$
190 LOCATE P(1/5),I:PRINT " <"
210 LOCATE 0,0:PRINT CT
300 IF CT/25=INT(CT/25) THEN LET
P(2)=P(2)+INT(RND*20)-10
410 LOCATE L,I:PRINT "M"
```

▼ VIC 20

```
30 PRINT CHR$(147)
60 LET W=21
140 LET P(1)=INT(RND(1)*(W-4))+1
160 PRINT CHR$(147)
170 FOR I=1 TO 3
172 FOR J=1 TO 5
174 PRINT
180 NEXT J
190 PRINT LEFT$(L$,P(1)); " <";
RIGHT$(L$, (18-P(1)))
210 PRINT CHR$(19);CT
240 GET Z$
300 IF CT/25=INT(CT/25) THEN LET
P(2)=P(2)+INT(RND(1)*20)-10
375 PRINT CHR$(19)
395 PRINT CHR$(19)
410 PRINT TAB(L)"M"
```

Que acontece se você mudar o número da linha 100? E se tirar o sinal de menos da linha 120? Mude os dois 25 da linha 300 para outro número e veja se faz alguma diferença.



# Labirinto Assombrado

É um lugar assustador. Os corredores escuros não parecem levar a parte alguma. Quem sabe são mal-assombrados? Você só pode ver o que está à frente, e só pode se mover nesta direção. Pode virar para a esquerda ou para a direita, mas com isso apenas muda seu ponto de vista. O único meio de escapar é chegar à cruz que assinala a saída.

Epa! O lugar é mal-assombrado, sim! Os fantasmas são representados pela letra F. Se você passa perto de um deles, é transportado instantaneamente para outra parte do labirinto. Aqui estão as teclas que você pode usar: X para avançar, N para virar para a esquerda e M para virar para a direita.

5 CLEAR 200	
10 DIM E(70)	Cria uma matriz para o labirinto.
20 DIM VS(4,3)	Cria uma matriz para guardar a parte do labirinto que pode ser vista na tela.
30 DIM F(3)	Cria uma matriz para guardar uma linha da matriz acima.
40 LET WS=""	
50 LET WS=WS+"0000000000"	
60 LET WS=WS+"0111100110"	
70 LET WS=WS+"0010011100"	
80 LET WS=WS+"0011010110"	
90 LET WS=WS+"0110100100"	
100 LET WS=WS+"0011111100"	
110 LET WS=WS+"0000009000"	Dados para o labirinto: 1 = corredor, 0 = parede, 9 = saída.
120 FOR I=1 TO 70	
130 LET E(I)=VAL(MID\$(WS,I,1))	"Loop" para guardar os dados na matriz E.
140 NEXT I	
150 LET S=-1	S é o número de vezes que o fantasma se moveu.
160 LET G=12	G é a posição do fantasma.
170 LET X=INT(RND(0)*50)+10	Escolhe uma posição inicial para o jogador.
180 IF E(X)<>1 THEN GOTO 170	Se a posição não fica em um corredor, volta para escolher outra.
190 GOSUB 860	Vai para a sub-rotina que posiciona o fantasma.
200 LET D=INT(RND(0)*4+1)	Escolhe a direção para onde o jogador está voltado no início do jogo.
210 IF X=G+10 OR X=G-10 THEN GOTO 170	
220 IF X=G+1 OR X=G-1 THEN GOTO 170	Verifica se o jogador está ao lado de um fantasma; em caso afirmativo, escolhe uma nova posição para o jogador.
230 LET H=H+1	Incrementa o número de movimentos do jogador.
240 IF H=5 THEN GOSUB 860	A cada 5 movimentos, vai para a sub-rotina que posiciona o fantasma.
250 GOSUB 390	Vai para a sub-rotina que imprime a parte do labirinto que o jogador pode ver.
260 LET AS=INKEY\$	
270 IF AS="" THEN GOTO 260	
280 IF AS="M" THEN LET D=D+1	
290 IF AS="N" THEN LET D=D-1	
300 IF D=5 THEN LET D=1	
310 IF D=0 THEN LET D=4	Recebe uma instrução do jogador e muda o valor de D, se necessário.
320 IF AS<>"X" THEN GOTO 210	Se a instrução não é "em frente", volta para 210.
330 IF D=1 AND E(X-10)<>0 THEN LET X=X-10	
340 IF D=3 AND E(X+10)<>0 THEN LET X=X+10	
350 IF D=2 AND E(X+1)<>0 THEN LET X=X+1	Verifica se é possível ir em frente.
360 IF D=4 AND E(X-1)<>0 THEN LET X=X-1	
370 IF E(X)=9 THEN GOTO 930	Verifica se o jogador chegou à saída.
380 GOTO 210	Se não chegou, volta para 210.



## Efeitos sonoros

Se o seu micro pode gerar sons, veja se consegue incluir um ruído assustador cada vez que um fantasma aparecer.

```

390 FOR I=1 TO 4
400 LET T=1-1
▼ 410 ON D GOTO 420,460,500,540
420 LET F(1)=X-10*T+1
430 LET F(2)=X-10*T
440 LET F(3)=X-10*T-1
450 GOTO 570
460 LET F(1)=X+10+T
470 LET F(2)=X+T
480 LET F(3)=X-10+T
490 GOTO 570
500 LET F(1)=X+10*T-1
510 LET F(2)=X+10*T
520 LET F(3)=X+10*T+1
530 GOTO 570
540 LET F(1)=X-T-10
550 LET F(2)=X-T
560 LET F(3)=X-T+10
570 FOR J=1 TO 3
580 IF F(J)<1 OR F(J)>69 THEN GOTO 840
590 IF E(F(J))=0 THEN LET V$(1,J)="H"
600 IF E(F(J))=1 THEN LET V$(1,J)=" "
610 IF E(F(J))=9 THEN LET V$(1,J)="+"
620 IF E(F(J))=2 THEN LET V$(1,J)="F"
630 NEXT J
640 NEXT I
650 LET V$(1,2)="Y"
▲● 660 CLS
670 PRINT
680 PRINT "**** LABIRINTO ****"
690 PRINT
700 PRINT "EM FRENTE   X"
710 PRINT "DIREITA      M"
720 PRINT "ESQUERDA     N"
730 FOR I=1 TO 5
740 PRINT
750 NEXT I
760 FOR I=4 TO 1 STEP-1
770 LET P$=""
780 FOR J=3 TO 1 STEP-1
790 LET P$=P$+V$(1,J)
800 NEXT J
810 PRINT TAB(7);P$
820 NEXT I
830 RETURN
840 LET V$(1,J)="H"
850 GOTO 630
860 LET E(G)=1
◆▼▲● 870 LET G=INT(RND(0)*50)+10
880 IF E(G)<1 THEN GOTO 870
890 LET E(G)=2
900 LET H=0
910 LET S=S+1
920 RETURN
930 PRINT "VOCE ESCAPOU"
940 PRINT "EM ";S*5+H;" MOVIMENTOS"
950 STOP

```

A listagem à esquerda é para o TRS-80. Para os outros micros, faça as mudanças abaixo.

○◆▼▲● Suprimir a linha 5  
▼ 130 LET E(1)=VAL(W\$(1 TO 1))  
▲● 170 LET X=INT(RND(1)\*50)+10  
◆▼ 170 LET X=INT(RND\*50)+10  
▲● 200 LET D=INT(RND(1)\*4)+1  
◆▼ 200 LET D=INT(RND\*4)+1  
● 260 AS="":IF PEEK(-16384)>127 THEN GET AS  
▲ 260 GET AS  
▼ 410 GOTO 380+40\*D  
● 660 HOME  
▲ 660 PRINT CHR\$(147)  
▲● 870 LET G=INT(RND(1)\*50)+10  
◆▼ 870 LET G=INT(RND\*50)+10

## Caracteres gráficos

Substitua os símbolos +, F, Y e # por caracteres gráficos. Para isso, terá que mudar as linhas 590, 610, 620, 650 e 840.

Sub-rotina que calcula:

- quais os quadrados que você pode ver de onde está.
- que caracteres colocar nesses quadrados.
- imprime instruções e a parte do labirinto que você pode ver.

## Aumente o labirinto

Este labirinto se baseia em um retângulo de 10x7 posições. Para aumentá-lo, basta mudar as seguintes linhas, substituindo N pelo número total de posições do novo retângulo (N deve ser múltiplo de 10).

10: Substitua 70 por N.

50-110: Acrescente novas linhas de "uns" e "zeros" em 55, 65 etc., de modo que o número total de caracteres seja N.

120: Substitua 70 por N.

170 e 870: Substitua 50 por N-20.

580: Substitua 69 por N-1.

Sub-rotina para mudar a posição do fantasma a cada cinco movimentos.

Mensagem no final do jogo.



# Sessão Espírita

As mensagens dos espíritos estão chegando, letra por letra. Eles querem que você observe as letras e entre com elas no computador, na ordem correta. Se você se enganar, os espíritos vão ficar zangados, muito zangados...

Preste atenção à estrela que aparece na tela. Ela lhe mostrará as letras que os espíritos estão enviando.

## Como o programa funciona

Depois de jogar algumas vezes, examine a listagem e veja se descobre como o programa funciona. (Pista: D é uma variável que diz ao computador em que lado do quadrado — 1, 2, 3, ou 4 — está a letra seguinte da mensagem.)

```

10 LET S=0
20 LET G=0
30 LET CS=64
40 CLS
50 PRINT
60 PRINT TAB(8); "SESSAO ESPIRITA"
70 FOR I=1 TO 8
80 LET X=6+I
90 LET Y=5
100 LET AS=CHR$(CS+1)
110 GOSUB 710
120 LET Y=11
130 LET AS=CHR$(CS+22-1)
140 GOSUB 710
150 NEXT I
160 FOR I=1 TO 5
170 LET X=5
180 LET Y=5+I
190 LET AS=CHR$(CS+27-1)
200 GOSUB 710
210 LET X=16
220 LET AS=CHR$(CS+8+1)
230 GOSUB 710
240 NEXT I
250 LET PS=""
260 LET N=INT(RND(0)*4+3)
270 FOR I=1 TO N
280 LET AS="*"
290 LET L=INT(RND(0)*26+1)
300 LET SS=CHR$(CS+L)
310 LET PS=PS+SS
320 LET D=4
330 IF L<22 THEN LET D=3
340 IF L<14 THEN LET D=2
350 IF L<9 THEN LET D=1
360 ON D GOTO 370,400,430,460
370 LET Y=6
380 LET X=L+6
390 GOTO 480
400 LET X=15
410 LET Y=L-3
420 GOTO 480
430 LET Y=10
440 LET X=28-L
450 GOTO 480
460 LET X=6
470 Y=32-L
480 GOSUB 710

```

```

490 FOR T=1 TO 600
500 NEXT T
510 LET AS=""
520 GOSUB 710
530 NEXT I
540 LET AS=""
550 LET X=0
560 LET Y=13
570 GOSUB 710
580 INPUT RS
590 IF RS=PS THEN GOTO 670
600 LET G=G+1
610 IF G=1 THEN PRINT "A MESA
    COMECOU A TREMER"
620 IF G=2 THEN PRINT "A LAMPADA
    ESTOUROU"
630 IF G=3 THEN GOTO 730
640 FOR T=1 TO 1000
650 NEXT T
660 GOTO 40
670 LET S=S+N
680 IF S<50 THEN GOTO 40
690 PRINT "OS ESPIRITOS SE FORAM"
700 STOP
710 PRINT@64*Y+2*X,AS;
720 RETURN
730 PRINT "UM PAR DE MAOS PEGAJOSAS
    APERTO O SEU PESCOCO"
740 STOP

```

A listagem acima é para o TRS-80. Para os outros micros, faça as mudanças abaixo.

```

40 HOME
40 PRINT CHR$(147)
260 LET N=INT(RND(1)*4+3)
260 LET N=INT(RND*4+3)
290 LET L=INT(RND(1)*26+1)
290 LET L=INT(RND*26+1)
360 GOTO 340+30*D
490 FOR T=1 TO 100
640 FOR T=1 TO 300
710 PRINT@32*Y+2*X,AS;
710 VTAB(Y):=HTAB(X+1):PRINT AS
710 PRINT AT Y,X;AS;
710 LOCATE X,Y:PRINT AS
710 PRINT CHR$(19):FOR K=1 TO Y:PRINT:
    NEXT:PRINT TAB(X);AS

```

O número da linha 490 controla a velocidade com que a estrela se move na tela. Quanto menor o número, maior a velocidade.



# Respostas dos problemas

## Pesadelo (página 3)

```

100 LET NS=CHR$(INT(RND(0)*26+65))
110 PRINT TAB(5);NS
230 IF FS(>)NS THEN GOTO 250
240 LET S=S+10+(ASC(NS)-65)

100 LET NS=CHR$(INT(RND(1)*26+65))
100 LET NS=CHR$(INT(RND*26+65))
240 LET S=S+10+(CODE(NS)-65)
    
```

Escolhe uma letra e a mostra na tela.  
Verifica se a tecla apertada está certa.  
Aumenta a contagem, que depende da letra (assim, por exemplo, Z vale mais que A).

Repare nas diferentes versões para diferentes micros.

## O Mago dos Números (página 5)

```

5 LET L=0
6 LET W=0
355 LET L=L+1
360 GOTO 390
380 LET W=W+1
390 PRINT
400 PRINT "VOCE O MAGO"
410 PRINT
420 PRINT " ";W;" ";L
430 FOR Q=1 TO 1000
440 NEXT Q
450 GOTO 20
    
```

Inicializa o número de vitórias e derrotas.  
Aumenta o número de vezes que você perdeu.  
Aumenta o número de vezes que você ganhou.  
Imprime a contagem.  
Pausa para que você possa ver a contagem.  
Volta para outra tentativa.

Aumente este número nos micros mais velozes.

## Mulher-Aranha (página 9)

```

5 LET NG=0
355 LET NG=NG+1
360 IF GS=TS THEN GOTO 370
362 PRINT "ERRADO!"
364 LET G=G+5
366 IF NG=2 THEN GOTO 410
368 GOTO 230
    
```

Inicializa o número de palpites.  
Incrementa o número de palpites.  
Se o palpite está certo, encerra o jogo.  
Imprime mensagem dizendo que o jogador errou.  
Aumenta de 5 o número de tentativas como penalidade para um palpite errado.  
Se você errou dois palpites, o jogo termina.  
Volta para a tentativa seguinte.

**Nota especial para os usuários do TK-2000:**

Se o seu micro é um TK-2000, use as modificações indicadas para a família Apple, com uma única exceção: se a modificação for do tipo  
nnn AS="":IF PEEK(-16384))127 THEN GET AS  
onde nnn é o número da linha e AS é o nome de uma variável, substitua esta linha por  
nnn AS="":CALL -16397:IF PEEK(-16385))0 THEN GET AS  
e acrescente as seguintes linhas:

```

1 DATA 169,255,141,0,192,77
2 DATA 16,192,141,255,191,96
3 FOR I=-16397 TO -16386
4 READ A:POKE I,A:NEXT I
    
```

**Nota para os usuários do MSX: Em todos os programas, acrescente a linha:**

```
5 R=RND(-TIME)
```

**Nota para os usuários do PC: Em todos os programas, acrescente a linha:**

```
5 RANDOMIZE VAL(RIGHT$(TIME$,2))
```



# Guias Práticos de Microcomputadores



**TUDO O QUE VOCÊ PRECISA  
SABER SOBRE MICROCOMPUTADORES**

**e as maravilhas que eles podem fazer**



**NOVOS \* COLORIDOS \* FASCINANTES**

92362/3

**LUTECIA**